DSC5211C QUANTITATIVE RISK MANAGEMENT

Workshop 6

Fernando Oliveira
bizfmdo@nus.edu.sg

Introduction to GAMS (www.GAMS.com)

In the transportation problem, we are given the supplies at several plants and the demands at several markets for a single commodity. We are also given the unit costs of shipping the commodity from plants to markets.

The economic question is: how much shipment should there be between each plant and each market so as to minimize total transport cost?

The algebraic representation of this problem:

Indices:

i = plants

j = markets

Given Data:

a_i = supply of commodity of plant i (in cases)

b_j = demand for commodity at market j (cases)

c_{ij} = cost per unit shipment between plant i and market j ($/case)

Decision Variables:

x_{ij} = amount of commodity to ship from plant i to market j (cases), where $x_{ij} \geq 0$, for all i, j

Constraints:

Observe supply limit at plant i: $\sum_j x_{ij} \leq a_i,\ \textit{for all } i$ (cases)

Satisfy demand at market j: $\sum_i x_{ij} \geq b_j,\ \textit{for all } j$ (cases)

Objective Function:

Minimize $\sum_i \sum_j c_{ij} x_{ij}$ ($K)

As an instance of the transportation problem, suppose there are two canning plants and three markets, with the given data as follows. (This example is adapted from Dantzig, 1963)

	Shipping Distances			Supplies
	Markets			
Plants	New York	Chicago	Topeka	
Seattle	2.5	1.7	1.8	350
San Diego	2.5	1.8	1.4	600
Demands	325	300	275	

Shipping distances are in thousands of miles, and shipping costs are assumed to be $90.00 per case per thousand miles.

The GAMS representation of this problem is as follows:

```
Sets
      i  canning plants   / Seattle, San-Diego /
      j  markets          / New-York, Chicago, Topeka / ;

Parameters
      a(i) capacity of plant i in cases
            /     Seattle      350
                  San-Diego    600 /

      b(j) demand at market j in cases
            /     New-York     325
                  Chicago      300
                  Topeka       275 / ;

Table d(i,j) distance in thousands of miles
                  New-York     Chicago      Topeka
      Seattle       2.5          1.7          1.8
      San-Diego     2.5          1.8          1.4 ;

Scalar f freight in dollars per case per thousand miles /90/ ;

Parameter c(i,j) transport cost in 1000s of dollars per case ;

      c(i,j) = f*d(i,j)/1000 ;

Variables
      x(i,j) shipment quantities in cases
      z      total transportation costs in 1000s of dollars ;

Positive variable x ;
```

(Continued in the next page…)

```
Equations
      cost          define objective function
      supply(i)     observe supply limit at plant i
      demand(j)     satisfy demand at market j ;

cost..     z =e= sum((i,j), c(i,j)*x(i,j)) ;

supply(i) ..  sum(j, x(i,j)) =l= a(i) ;

demand(j) .. sum(i, x(i,j)) =g= b(j) ;

Model transport /all/ ;

solve transport using lp minimizing z ;
display x.l, x.m ;
```

Structure of a GAMS model:

Inputs

Sets
- Declaration
- Assignment of members

Data (Parameters, Tables, Scalars)
- Declaration
- Assignment of Values

Variables
- Declaration
- Assignment of type

Assignment of bounds and/or initial values for the variables (optional)

Equations
- Declaration
- Definition

Model and Solve Statements

Display Statement (Optional)
Output to Files (Optional)

Running the Transport model:

Step 1: Start your GAMS
Step 2: Go to File/Model Library/ Open Gams model library
Step 3: Sort problems by SeqNr
Step 4: Open problem number 001 - Transport

Notes:

1. $Title: used to insert title of the programme
2. $Ontext

Text and Comments Go here
$Offtext

3. **********You can insert comments using *

```
$Title  A Transportation Problem (TRNSPORT,SEQ=1)
$Ontext

This problem finds a least cost shipping schedule that meets
requirements at markets and supplies at factories.


Dantzig, G B, Chapter 3.3. In Linear Programming and Extensions.
Princeton University Press, Princeton, New Jersey, 1963.

This formulation is described in detail in:
Rosenthal, R E, Chapter 2: A GAMS Tutorial. In GAMS: A User's Guide.
The Scientific Press, Redwood City, California, 1988.

The line numbers will not match those in the book because of these
comments.

$Offtext
```

a) Analyze the structure of the transport model. Write down the equations.
b) Run the model.

You can do this by using:
File/Run
Or F9
Or clicking on .

c) Analyze the outputs and interpret your solution.

Go to file transport.lst
Click on Display

Interpret the meaning of X.L and X.M

d) Change the parameters in the model and test how they affect the results. Discuss your results with your colleagues.